CS 545: Data & Information Visualization

Assignment 3: Juxtaposed Interactive Choropleth Maps
Group 6: Nathan Chapman, Andrew Struthers, Kollin Trujillo, Gihanen, Nick Haviland
Date: Thu 9 Feb 2023

Importing and Parsing Census Data

In[31]:=
```
Clear[dataEncoded, itemCodeList, dataDecoded]
(*Importing data and parsing into dataset*)
dataEncoded = Import[
   "https://www2.census.gov/programs-surveys/stc/datasets/2021/FY2021-Flat-File.txt",
   {"CSV", "Dataset"}, "HeaderLines" → {1, 1}];
(*Import and parse item codes*)
itemCodeList = Rule @@@ Import[
     "https://www2.census.gov/programs-surveys/stc/technical-documentation/file-layout/
       taxitemcodesanddescriptions.xls", "Data"][[1, 2 ;;]];
(*Replace the item codes with their actual names*)
dataDecoded = Normal@Normal@dataEncoded /. Append[itemCodeList,
     "X" → "X"(*Missing value replacement*)] // Association // Dataset
```

Out[34]=

	US	AL	AK
Property Taxes	23 330 874	468 015	120 0
General Sales and Gross Receipts Taxes	370 907 383	3 912 037	X
Alcoholic Beverages Sales Tax	7 557 257	272 021	41 12
Amusements Sales Tax	8 643 203	0	15 48
Insurance Premiums Sales Tax	26 293 455	468 162	61 55
Motor Fuels Sales Tax	51 477 011	878 038	44 75
Pari–mutuels Sales Tax	156 610	4112	X
Public Utilities Sales Tax	12 178 343	663 891	4361
Tobacco Products Sales Tax	19 142 532	170 998	61 18
Other Selective Sales and Gross Receipts Taxes	54 362 034	645 263	39 38
Alcoholic Beverages License	751 236	3667	1279
Amusements License	592 304	X	X
Corporations in General License	9 500 800	187 666	X
Hunting and Fishing License	1 918 283	21 506	32 96
Motor Vehicles License	30 377 196	260 114	36 47
Motor Vehicle Operators License	3 130 082	43 783	X
Public Utilities License	1 083 682	19 815	7835
Occupation and Businesses License, NEC	15 697 004	69 799	74 62
Other License Taxes	1 841 279	1	0
Individual Income Taxes	506 264 685	4 908 837	X

rows 1–20 of **25** columns 1–10 of **52**

In[35]:=
```
(*Choose the 8 datasets with the fewest missing values, then randomly reorder them*)
Clear@dataCore
dataCore = SortBy[dataDecoded, Count["X"]][[ ;; 8]] // RandomSample;
(*How many missing values are in the core data set*)
"There are " <> ToString@Count[dataCore, "X", ∞] <> " missing data points"
(*What are the items in the core data set*)itemList = Normal@Keys@dataCore;
```

Out[37]=
```
There are 0 missing data points
```

Choropleth

In[39]:=
```
(*Get state entities to be used in the choropleth*)
stateList = GeoEntities[Entity["Country", "UnitedStates"], "USState"];
```

In[63]:=
```
Clear@choropleth
choropleth[item_String, Optional["UnitedStatesContinental" → q : (True | False),
    "UnitedStatesContinental" → False]] := Rasterize[
  GeoRegionValuePlot[
   # → dataCore[item][#["StateAbbreviation"]] & /@
    If[q, DeleteCases[state_ /; state["StateAbbreviation"] == "AK" ||
          state["StateAbbreviation"] == "HI"], Identity]@stateList,
   PlotLabel → item,
   MissingStyle → Green,
   TargetUnits → "US kDollars",
   GeoLabels → (Tooltip[#1, #2["StateAbbreviation"] <> ": $" <> ToString@#4] &)(*,
   ImageSize→Medium*)
  ],
  ImageResolution → 300
 ]
choropleth[y_][x_] := choropleth[x, y]
```

In[66]:=
```
(*Test choropleth juxtaposition*)
choropleth["UnitedStatesContinental" → True] /@
  {"Motor Fuels Sales Tax", "Motor Vehicles License"};
GraphicsRow[%, 0, ImageSize → 72 * 8.5]
```

Out[67]=



Interactive Comparison

In[68]:=
```
Manipulate[
 GraphicsRow[
  Show[choropleth["UnitedStatesContinental" → cont]@#, ImageSize → Large] & /@
   {item1, item2},
  0,
  ImageSize → 72 * 8.5
 ],
 {{cont, False, "Continental US"}, {True, False}},
 {{item1, itemList[[1]], "Left Dataset"}, itemList[[ ;; 4]]},
 {{item2, itemList[[5]], "Right Dataset"}, itemList[[5 ;;]]}
]
```

Out[68]=